

# JD-számító program

Szinte minden csillagászati észlelőprogram lelke az időkülönbségek gyors és pontos átszámítása. Az egy napon belüli időpontok átszámítása nem ütközik különösebb problémába, de ez nagyon ritkán fordul elő csillagászati programok esetében. A nehézség ott kezdődik, amikor két időpont között eltelt napok számát kell kiszámítanunk. Ilyen probléma a csillagászati átszámításokban már sokkal többször előfordul. Például, ha egy változócsillagnak meghatároztuk a maximumidőpontjait, és kíváncsiak vagyunk az átlagperiódusára. De számtalan ilyen példát említhetnék. Ez legjobban a Julián Dátum használatával oldható meg.

Ezt az időszámítási formát Scaliger javasolta 1582-ben, azonban elsőként csak John Herschel használta észleléseinek feljegyzéséhez. A Julián Dátum nem más, mint a napok egymás utáni számolása minden struktúra nélkül. Vagyis minden nap kap egy sorszámot egy kiválasztott időponttól. Scaliger ezt a kiválasztott időpontot időszámításunk előtt 4713 január 1-jén, déli 12 órára tette. Mivel a napok váltása délben történik, így az észlelőnek nem kell az órát figyelni, hogy elmúlt-e már éjfél. A napot pedig tizednapokra osztották fel. Így például az éjfél 0,5-nek felel meg.

Egy ilyen kis program megírása nem okozhat problémát egy programozásban jártas amatőrcsillagásznak. Ezért gondolom, hogy már sokan használnak ilyen konverziós programot. Ám legtöbbször ezek a programok csak egy bizonyos időintervallumon belül számolnak jól. A másik probléma a Julianus- és a Gergely-naptár közötti különbség. Ha mindezeket a feltételeket valaki bele szeretné venni a programjába, akkor már nem is olyan egyszerű a probléma. Ezért éreztem úgy, hogy időszerű lenne egy univerzális program közzlése, amiből mindenki azt a részt hagyja el, ami számára nem szükséges.

Egy ilyen programot találtam a Tudomány és technika Commodore 64-re című Data Becker könyv 12. fejezetében. A másik ilyen program, amit ebben a cikkben is közlünk, a Sky and Telescope 1984 májusi számában található.

```
100 REM  DATUM  -->  JD
105 REM
110 INPUT "EV,HONAP,NAP ":Y,M,D
115 INPUT "JULIAN VAGY GERGELY NAPTAR (0/1) ":G
120 D1=INT(D):F=D-D1-0.5
125 J=-INT(7*(INT((M+9)/12)+Y)/4)
130 IF G=0 THEN 150
135 S=SGN(M-9):A=ABS(M-9)
140 J1=INT(Y+S*INT(A/7))
145 J1=-INT((INT(J1/100)+1)*3/4)
150 J=J+INT(275*M/9)+D1+G*J1
155 J=J+1721027!+2*G+367*Y
160 IF F>=0 THEN 170
165 F=F+1:J=J-1
170 PRINT "J.D. : ";J,F
175 END

200 REM  JD  -->  DATUM
205 REM
```

```

210 INPUT "J.D. ";J,F
215 INPUT "JULIAN VAGY GERGELY NAPTAR (0/1) ";G
220 F=F+0.5
225 IF F<1 THEN 235
230 F=F-1:J=J+1
235 IF G=1 THEN 245
240 A=J:GOTO 255
245 A1=INT((J/36524.25)-51.12264)
250 A=J+1+A1-INT(A1/4)
255 B=A+1524
260 C=INT((B/365.25)-0.3343)
265 D=INT(365.25*C)
270 E=INT((B-D)/30.61)
275 D=B-D-INT(30.61*E)+F
280 M=E-1:Y=C-4716
285 IF E>13.5 THEN M=M-12
290 IF M<2.5 THEN Y=Y+1
295 PRINT "DATUM : ";Y,M,D
300 END

```

Az első program futása a következőképpen néz ki:

```

RUN
EV,HONAP,NAP ? 1989,4,10.32
JULIAN VAGY GERGELY NAPTAR (0/1) ? 1
J.D.      2447626      .82

```

READY

A második programot GOTO 200 paranccsal tudjuk indítani.

```

GOTO 200
J.D. ? 2447626..82
JULIAN VAGY GERGELY NAPTAR (0/1) ? 1
DATUM : 1989      4      10.32

```

READY

Az adatok beadásánál vigyázzunk a vessző és a tizedespont különbségére.

Végül megadunk néhány időpontot a programok teszteléséhez.

| Julián naptár | Gergely naptár | Julián dátum |
|---------------|----------------|--------------|
| -4712 1 1.5   | -4713 11 24.5  | 0.0          |
| - 584 5 28.6  | - 584 5 22.6   | 1507900.1    |
| + 200 3 1.0   | + 200 3 1.0    | 1794167.5    |
| +1990 4 5.5   | +1990 4 18.5   | 2448000.0    |
| +1999 12 19.5 | +2000 1 1.5    | 2451545.0    |
| +3000 2 29.9  | +3000 3 21.9   | 2816867.4    |

ZALEZSÁK TAMÁS